

Human-Nature Interaction in World Modeling with Modelica

Rodrigo Castro^{1,2}, Peter Fritzson³, François Cellier⁴, Safa Motesharrei⁵, Jorge Rivas⁶

¹Department of Environmental Systems Science
ETH Zurich, Switzerland

²Department of Computer Science, University of Buenos Aires
and CIFASIS-CONICET, Argentina

³Department of Computer and Information Science
Linköping University, SE-581 83 Linköping, Sweden

⁴Computer Science Department
ETH Zurich, Switzerland

⁵National Socio-Environmental Synthesis Center (SESYNC)
Annapolis, MD, 21401, USA

⁶Department of Political Science, University of Minnesota
Minneapolis, MN 55408, USA

rodrigo.castro@usys.ethz.ch, peter.fritzson@liu.se, fcellier@inf.ethz.ch,
ssm@umd.edu, jorgerodrigorivas@gmail.com

Abstract

It is our predicament that we live in a finite world, and yet we behave as if it were infinite. Steady exponential material growth with no limits on resource consumption and population is the dominant conceptual model used by today's decision makers. This is an approximation of reality that is no longer accurate and started to break down. The World3 model, originally developed in the 1970s, includes many rather detailed aspects of human society and its interaction with a resource-limited planet. However, World3 is a rather complex model. Therefore it is valuable for pedagogical reasons to show how similar behavior can be also realized with models that are much simpler. This paper presents a series of world models, starting with very simple exponential growth and predator-prey systems, then investigates a minimal human-nature model, Handy, and ends with a brief account of the World3 model. For the first time, a simple human-nature interaction model is made available in Modelica that distinguishes between dynamics of Elite and Commoner social groups. It is shown that Handy can reproduce rather complex behavior with a very simple model structure, as compared to that of world models like World3.

Keywords: Modelica, Simulation, Population Dynamics, World Models, Human-Nature Models.

1 Introduction

Dynamic modeling can be applied to the human population and its interaction with the earth system. Certain types of such models are called World models.

Perhaps the earliest study indicating that human population would eventually reach limits to growth is that presented by Richard Malthus in 1798 in his famous work *An Essay on the Principle of Population* [4]. A catastrophe in the form of population collapse was considered as a possible consequence of uncontrolled growth that would hit the limits of food production causing massive famine.

A series of simulation models were developed mainly during the 1970's, with some later updates, aiming at understanding the complexity of the interactions between global societies and their physical environment, searching for the conditions that would lead to a collapse and possible measures that can avoid it. They formed the category of so-called World Models, typically of considerable generality and complexity, spanning several subsystems (demography, energy, economy, industry, agriculture, etc.) at varied levels of detail.

2 Growth without Limitations

In this first example, we shall show the dynamics of human populations in an idealized ecological system without limitations. We assume that a population contains P individuals. In general, the rate of change of the number of individuals is the difference between the population birth rate and its death rate:

$$\dot{P} = \text{birthrate} - \text{deathrate} \quad (1)$$

where \dot{P} means dP/dt , the rate of change of the population stock. It is natural to assume that the birth and death rates are proportional to the size of the population:

$$\begin{aligned} \text{birthrate} &= g \cdot P \\ \text{deathrate} &= d \cdot P \end{aligned} \quad (2)$$

where g is the growth (birth) factor and d the decrease (death) factor for the population. These factors can be assumed to be constant if the same amount of food per individual is available independent of the size of the population, and if no sudden lethal events (disease, war) affect the population massively. These assumptions are of course valid only for a reasonably limited size of the population, since food supply is never infinite in a closed system and epidemics, pandemics, and wars sometimes break out. Putting (1) and (2) together gives:

$$\dot{P} = (g - d) \cdot P \quad (3)$$

Solutions to this equation yield an exponentially increasing population if $(g - d) > 0$ or an exponentially decreasing one if $(g - d) < 0$. We represent this equation in the model class `PopulationGrowth` below:

```
class PopulationGrowth
  parameter Real g = 0.04 "Growth factor";
  parameter Real d = 0.0005 "Death factor";
  Real P(start=10) "Pop. size, init 10";
equation
  der(P) = (g-d)*P;
end PopulationGrowth;
```

The model is simulated and the population size P is plotted below:




Figure 1. Exponential growth of a population with unlimited food supply.

As expected, Figure 1 shows an exponentially increasing population. We should, however, remember that exponential growth can never persist indefinitely in a closed system with a limited food supply.

3 A Predator-Prey Model with Limitations in Prey Animal Population

Now we will study a more interesting system consisting of interacting populations of predators and prey animals.

The simplest predator-prey model, the so-called Lotka-Volterra model [3], consists of two population stocks. We may think of predators as foxes and preys as rabbits.

The rabbit population of size R is assumed to have an unlimited food supply. Therefore equation (3) applies regarding the birth rate of the rabbit population, with a positive growth term $g_r \cdot R$, where g_r is the growth factor for rabbits.

On the other hand, the fox population of size F feeds on the rabbits. The rabbit death rate can be assumed to be proportional to the number of foxes due to increased hunting pressure, and to the number of rabbits due to the higher probability of success in hunting, giving a rabbit death rate due to foxes of $d_{rf} \cdot F \cdot R$, where d_{rf} is the death factor of rabbits due to foxes. Putting the birth and death rates together gives the total rabbit population change rate \dot{R} as defined by equation (4) below:

$$\dot{R} = g_r \cdot R - d_{rf} \cdot F \cdot R \quad (4)$$

The growth of the fox population of size F is proportional to the rate of rabbits consumed, i.e., those that die because they were hunted by foxes, which is the death rate term $d_{rf} \cdot F \cdot R$ mentioned in the previous equation. The efficiency of growing foxes from rabbits is determined by the growth factor g_{fr} , giving a fox population growth rate term $g_{fr} \cdot d_{rf} \cdot R \cdot F$. The fox population also has an intrinsic death rate $d_f \cdot F$ proportional to the size of the fox population by the fox death factor d_f . By combining these two terms, we obtain equation (5):

$$\dot{F} = g_{fr} \cdot d_{rf} \cdot R \cdot F - d_f \cdot F \quad (5)$$

where \dot{F} is the total change rate of the fox population.

We have created a Modelica [2][8] class below called `LotkaVolterra` that includes both equations (4) and (5) along with all the variables and constant proportionality factors. The two state variables R and F for the sizes of rabbit and fox populations are represented by the Modelica variables `rabbits` and `foxes`:

```

class LotkaVolterra
  parameter Real g_r =0.04    "Natural
  growth rate for rabbits";
  parameter Real d_rf=0.0005 "Death rate
  of rabbits due to foxes";
  parameter Real d_f =0.09    "Natural
  deathrate for foxes";
  parameter Real g_fr=0.1    "Efficency
  in growing foxes from rabbits";
  Real rabbits(start=700) "Rabbits, (R)";
  Real foxes(start=10)    "Foxes, (F)";
equation
  der(rabbits) = g_r*rabbits -
    d_rf*rabbits*foxes;
  der(foxes) = g_fr*d_rf*rabbits*foxes -
    d_f*foxes;
end LotkaVolterra;

```

The model is simulated, and the sizes of the rabbit and fox populations as a function of time are shown below (Figure 2):




Figure 2. Number rabbits—prey animals, and foxes—predators, as a function of time simulated from the PredatorPrey model.

The Lotka-Volterra model has a rather special property: the solution variables, i.e., the rabbit and fox population sizes, do not approach constant steady-state levels (unless they start at those equilibrium levels, which is rarely the case). Instead, they exhibit oscillatory equilibrium. The shape of these oscillations, very characteristic for Lotka-Volterra models, corresponds rather well to several experimental population studies of predators and prey animals. However, this does not prove these partly inductive model equations to be correct, only that they seem to have certain approximate physical significance.

4 A Minimal Human-Nature Dynamical Model with Economic Stratification

Simplified approaches to World models can exhibit interesting behavioral patterns while keeping model complexity low enough to enable intuitive comprehension. One such simplified model is Handy (Human and Nature Dynamical model) [9] developed jointly by researchers at the University of Maryland and the University of Minnesota¹.

The LotkaVolterra model (Section 3) was the original inspiration behind Handy, which is not a world model in the typical sense, although, in principle, it can be applied to a homogenous world.

Handy has four differential equations describing the evolution of its state variables: Commoner population (*commoners*), Elite population (*elite*), regenerating natural resources (*nature*), and accumulated wealth (*wealth*). Human population plays a role analogous to that of predators, and nature plays the role of the resource preyed upon.

An interesting feature of Handy is that it introduces the accumulation of economic *wealth*, and divides the human population into rich and poor according to their unequal access to available wealth.

This new variable explains why human societies can undergo an irreversible collapse, while animal populations show cyclic changes (or reversible collapses).

Social inequality is not only explicitly considered but also plays a key role in the sustainability analyses of the model. This makes Handy the first model of its kind that studies the impacts of inequality on the fate of societies, a capability seldom found even in complex world models.

Handy establishes a useful general framework that allows carrying out “thought experiments” about societal collapse scenarios and the changes that might avoid them.

The model is a very strong simplification of the human-nature system, which results in many limitations. Despite its simplicity, such a model is easy to understand and offers a more intuitive grasp of underlying dynamical phenomena compared to more complex and less aggregated models.

We briefly introduce a much more advanced class of world model called `World3` in Section 5 of this paper. This model (the compiled version) features 41 state variables and 245 algebraic variables. It captures many

¹ The Handy model has been publicly available since 2010 in the form of unpublished reports and presentations, deposited on the web site of its authors:

http://www.atmos.umd.edu/~ekalnay/#publications_based.

aspects of the human system and ecology, and it has been calibrated against empirical data.

4.1 Handy Model Equations, Description, and Modeled Societies

The Handy model divides the total population into two classes: Commoners and Elites, of population sizes `commoners` and `elite`, respectively. Population grows through birth rates `birthRComm` and `birthRElite`, which are constants. The population shrinks through death rates `deathRComm` and `deathRElite`, which are modeled as functions of `wealth`.

The state variables of the model are described in Table 1, additional variables in Table 2, and the parameters of the model in Table 3.

A single stock (variable), `nature` represents an amalgamation of nonrenewable and renewable resources, including “regenerating” resources such as forests, soils, and animal and fish stocks, etc.

The dynamics of the model are defined by the following four differential equations:

$$\begin{aligned} \text{commoners}' &= \text{birthRComm} \cdot \text{commoners} - \text{deathRComm} \cdot \text{commoners} \\ \text{elite}' &= \text{birthRElite} \cdot \text{elite} - \text{deathRElite} \cdot \text{elite} \\ \text{nature}' &= \text{natureRegen} \cdot \text{nature} \cdot (\text{natureCap} - \text{nature}) \\ &\quad - \text{depletFactor} \cdot \text{commoners} \cdot \text{nature} \\ \text{wealth}' &= \text{depletFactor} \cdot \text{commoners} \cdot \text{nature} \\ &\quad - \text{consRComm} - \text{consRElite} \end{aligned}$$

which in Modelica are expressed as:

```
der(commoners) = birthRComm*commoners -
deathRComm*commoners;
der(elite) = birthRElite*elite -
deathRElite*elite;
der(nature) =
natureRegen*nature*(natureCap-nature) -
depletFactor*commoners*nature;
der(wealth) =
depletFactor*commoners*nature - consRComm-
consRElite;
```

The equation for `nature` includes a regeneration term `natureRegen*nature*(natureCap-nature)` and a depletion term `depletFactor*commoners*nature` intended to also include degradation in nature caused by pollution.

The regeneration term is synthetic (i.e., not directly physical) following an s-shaped form² parameterized to resemble physically realistic results.

It exhibits exponential growth for low values of `nature` (because `natureCap-nature` is almost constant for small `nature`), reaches its maximum at `nature=natureCap/2`, and becomes small when `na-`

`nature` approaches `natureCap` (maximum size of `nature` in absence of depletion).

The depletion term includes a rate of depletion (pollution) per worker `depletFactor` making it proportional to both `nature` and `commoners` (labor). The economic activity of Elite is modeled to represent supervisory functions with no direct influence on the extraction of resources or the production of wealth. The underlying concept is that “only commoners produce.” Handy does not model the impacts of technological change, considering that technology through history has proven to produce both increases and decreases in resource use efficiency. Thus, a simplification is made assuming that these effects cancel each other out.

Accumulated Wealth (`wealth`) increases with production (`depletFactor*commoners*nature`) and decreases according to the consumption rates of the Elites and the Commoners, `consRComm` and `consRElite`, respectively.

Table 1. Main state variables of the Handy model.

Name	Description	Unit	Typical initial value(s)
commoners	Commoner population size	Number of people	100
elite	Elite population size	Number of people	0, 1, 25
nature	Nature's stock	EcoDollars	<code>natureCap</code> = 100
wealth	Accumulated wealth	EcoDollars	0

The additional variables of the model are described in Table 2. Two of these variables describe consumption. The consumption of the Commoners, `consRComm`, is given by a subsistence salary per capita `subsSal` times the working population. Elites earn a salary that is `ineqFactor` times larger than that of Commoners, intended to simulate class inequality, and is considered time-independent for any given scenario. This is reflected in the following equations:

$$\begin{aligned} \text{consRComm} &= \min(1, \text{wealth} / \text{wealthMin}) \cdot \text{subsSal} \cdot \text{commoners} \\ \text{consRElite} &= \min(1, \text{wealth} / \text{wealthMin}) \cdot \text{subsSal} \cdot \text{elite} \cdot \text{ineqFactor} \end{aligned}$$

which in Modelica are expressed as:

```
consRComm = min(1, wealth/wealthMin) *
subsSal*commoners;
consRElite = min(1, wealth/wealthMin) *
subsSal*elite*ineqFactor;
```

Both consumption rates, `consRComm` and `consRElite`, are subject to a “famine” minimum threshold for accumulated wealth before famine, `wealthMin`. The consumption rates are linearly reduced down to zero by the `min()` terms in the above equations, when `wealth` falls below `wealthMin`.

The death rates of the Commoners and the Elite are defined by the following equations:

² This is the well-known “logistic equation” used in many domains of life sciences to represent exponential growth followed by smooth saturation.

$$\begin{aligned}
 deathRComm &= deathRnormal + \max(0, 1 - consRComm) \\
 &\quad / (subsSal \cdot commoners) \cdot (deathRfamine - deathRnormal) \\
 deathRElite &= deathRnormal + \max(0, 1 - consRElite) \\
 &\quad / (subsSal \cdot elite) \cdot (deathRfamine - deathRnormal)
 \end{aligned}$$

The carrying capacity (`carryingCap`) of population, i.e., the population that can be sustained by the re-growth of nature, and its corresponding maximum carrying capacity (`carryingCapMax`), attainable under certain conditions, are defined by these equations:

$$\begin{aligned}
 carryingCap &= natureRegen / depleteFactor \cdot \\
 &\quad (natureCap - subsSal \cdot eta / depleteFactor)
 \end{aligned}$$

$$carryingCapMax = natureRegen / (eta \cdot subsSal) \cdot (natureCap / 2)^2$$

The threshold for minimum wealth (`wealthMin`), the total population, and a dimensionless quotient `eta` used above are defined as follows:

$$\begin{aligned}
 wealthMin &= consWorkerMin \cdot commoners + \\
 &\quad ineqFactor \cdot consWorkerMin \cdot elite \\
 eta &= (deathRfamine - birthRComm) / \\
 &\quad (deathRfamine - deathRnormal)
 \end{aligned}$$

$$totalPopulation = elite + commoners$$

Table 2. Additional variables of the Handy model for Human and Nature Dynamics.

Name	Description	Unit
<code>deathRComm</code>	Commoners' death rate	Number of people/year
<code>deathRElite</code>	Elites' death rate	Number of people/year
<code>consRComm</code>	Consumption rate of commoners	EcoDollars/year
<code>consRElite</code>	Consumption rate of Elites	EcoDollars/year
<code>eta</code>	Dimensionless quotient	
<code>carryingCap</code>	Carrying capacity (of the whole system)	Number of people
<code>carryingCapMax</code>	Maximum carrying capacity	Number of people
<code>totalPopulation</code>	Total population	Number of people

The parameters of the model are described in Table 3.

Table 3. Parameters of the Handy model for Human and Nature Dynamics.

Name	Description	Typical value(s)
<code>deathRnormal</code>	Healthy normal (minimum) death rate	0.01
<code>deathRfamine</code>	Famine (maximum) death rate	0.07
<code>birthRComm</code>	Commoner birth rate	0.03, or 0.065
<code>birthRElite</code>	Elite birth rate	0.03, or 0.02
<code>subsSal</code>	Subsistence salary per person	0.0005
<code>consWorkerMin</code>	Minimum required consumption per worker	0.005
<code>natureRegen</code>	Nature's regeneration rate	0.01

<code>natureCap</code>	Nature's carrying capacity	100
<code>ineqFactor</code>	Inequality factor between elite and commoners	1, 10, or 100
<code>depletFactor</code>	Depletion (production) factor	6.67E-6, or 6.35E-6, or 13E-6

There are three dimensions for quantities in the model: Population (either Commoners or Elites), in units of people, Nature or Wealth, in units of EcoDollars, and Time, in units of years. All other parameters and functions in the model carry units that are compatible with these dimensions.

All of the above definitions appear in the HandyBase Modelica model below, which is a partial model that is inherited by the following actual simulated models.

```

partial model HandyBase "Equations,
variables, and parameters for Handy"
  // All parameters, default values for
egalitarian society
  parameter Real birthRComm = 0.03 "Birth
Rate of Commoners No people/year";
  parameter Real birthRElite = 0.03 "Birth
Rate of Elite. No of people/yr";
  parameter Real natureRegen = 0.01
"Nature's regeneration factor.";
  parameter Real natureCap = 100.0
"Nature's capacity. ecoDollars";
  parameter Real subsSal=0.0005
"Subsistence Salary/Worker.
Dollars/worker";
  parameter Real consWorkerMin = 0.005
"Minimum required Consumption per
Worker. Dollars/worker";
  parameter Real depletFactorEq =
0.00000667
"Rate of depletion (pollution) per
worker at Equilibrium. 1/Worker*year";
  parameter Real deathRnormal = 0.01
"Healthy Death Rate. people/year";
  parameter Real deathRfamine = 0.07
"Famine Death Rate. people/year";
  parameter Real ineqFactor = 0
"Inequality in consumption level for
Workers and Non-Workers. Does not play a
role when elite=0";

  Real depletFactor "Rate of depletion per
worker. 1/Worker*year";
  Real commoners "Population of Commoners.
Number of people";
  Real elite "Population size of Elite.
Number of people";
  Real nature "Natural stock (renewable
and nonrenewable). ecoDollars";
  Real wealth "Accumulated wealth.
EcoDollars";
  Real wealthMin "Minimum threshold accum
wealth before famine. EcoDollars";
  Real deathRComm "Death Rate for
Commoners. Number of people/year";
  Real deathRElite "Death Rate for Elite.
Number of people/year";

```

```

Real consRComm "Consumption Rate of
Commoners. Dollars/year";
Real consRElite "Consumption Rate of
Elite. Dollars/year";
Real eta "Derived quotient expression.
Dimensionless";
Real carryingCap "Carrying Capacity.
Number of people";
Real carryingCapMax "Maximum Carrying
Capacity. Number of people";
Real totalPopulation "Total population:
elite and commoners. No people";

```

equation

```

der(commoners) = birthRComm*commoners -
deathRComm*commoners;
der(elite) = birthRElite*elite -
deathRElite*elite;
der(nature) = natureRegen*nature *
(natureCap-nature) -
depletFactor*commoners*nature;
der(wealth) = depletFactor* commoners *
nature - consRComm - consRElite;
deathRComm = deathRnormal + max(0,
1-consRComm / (subsSal*commoners)) *
(deathRfamine-deathRnormal);
deathRElite = deathRnormal + max(0,
1-consRElite / (subsSal*elite)) *
(deathRfamine-deathRnormal);

consRComm = min(1, wealth/wealthMin) *
subsSal*commoners;
consRElite = min(1, wealth/wealthMin) *
ineqFactor*subsSal*elite;
wealthMin = consWorkerMin*commoners +
ineqFactor*consWorkerMin*elite;

eta = (deathRfamine - birthRComm) /
(deathRfamine-deathRnormal);
carryingCap = natureRegen /depletFactor
*(natureCap - subsSal*eta/ depletFactor);
carryingCapMax = natureRegen /
(eta*subsSal) * (natureCap/2)^2;
totalPopulation = elite + commoners;

```

initial equation

```

nature = natureCap;
wealth = 0;
end HandyBase;

```

4.2 Types of Societies, Simulation Methodology and Scenarios

Handy has been applied to three types of societies:

- *Egalitarian society*—with no elites, i.e., the elite population `elite = 0`. Scenario models `HandyEgal1` to `ModelEgal4`.
- *Equitable society*—with workers (commoners) and elites (non-workers), where both groups earn the same per person, i.e., `ineqFactor=1`. Scenario models `HandyEquit1` to `HandyEquit5`.
- *Unequal society*—with commoners and elites, where elites consume more per capita than commoners, i.e., `ineqFactor > 1`. Scenario models `HandyUnEq1` to `HandyUnEq4`.

Several scenarios will be studied for each kind of society. The default values for parameters and initial values for state variables used in these scenarios are those presented in Table 1 and Table 3.

The thought experiments are performed as simulations, the results of which are presented in Sections 4.3 and 4.5, respectively. Within each type of society, different scenarios are studied by varying the rate of depletion per worker called `depletFactor` (short for depletion factor).

For each type of society, an equilibrium state can be reached, in which all derivatives go to zero, and consequently, all the system's variables settle into a steady state.

Under that condition, the carrying capacity can be maximized if nature's regeneration rate is maximal. According to the equation for nature, the latter is satisfied when `nature=natureCap/2`. The depletion factor `depletFactor` is at the optimal level `depletFactorEq`, when it produces a steady state with a maximum sustainable population.

The scenarios below explore the consequences of different types of societies as they behave optimally or sub-optimally, determined by the depletion factor.

4.3 Simulation Scenarios of an Egalitarian Society

For this type of society, the optimal value for the depletion factor that maximizes the carrying capacity (`carryingCap`) is `depletFactorEq=6.67E-6` (derived analytically). Below is the `HandyEgalitarianBase` model that is being inherited by all `Egal` models.

```

partial model HandyEgalitarianBase
"Egal - Scenarios with Egalitarian
Society (No-Elite)"
extends HandyBase;
initial equation
elite = 0; // For All "Egal" scenarios
commoners = 100.0;
end HandyEgalitarianBase;

```

Scenario Model HandyEgal1: "Soft landing to Equilibrium"

```

model HandyEgal1 "Scenario EGal1: Soft
Landing to Equilibrium"
extends HandyEgalitarianBase;
equation
depletFactor = depletFactorEq;
end HandyEgal1;

```

By making `depletFactor=depletFactorEq` the results in Figure 3 below are obtained:

```

plot({commoners_, elites_, nature_,
wealth_, carryingCap_, carryingCapMax_})

```




Figure 3. Soft landing to equilibrium, with optimal nature depletion factor. Simulation of HandyEgal1.

In this example, maximum nature regeneration can support a maximum sustainable depletion rate (pollution) and a steady maximum population (commons) equal to the carrying capacity (carryingCap).

Scenario Model HandyEgal3: “Cycles of Prosperity and Collapse”

In this scenario we increase the depletion intensity by selecting `depletFactor = 4 * depletFactorEq`.

```

model HandyEgal3 "Scenario Egal3: Cycles
  of Prosperity and Collapse"
  extends HandyEgalitarianBase;
  equation
    depletFactor = 4 * depletFactorEq;
  end HandyEgal3;
    
```

The results obtained are shown in Figure 4 below:

```

plot({commons_, elites_, nature_,
  wealth_, carryingCap_, carryingCapMax_})
    
```




Figure 4. Cycles of Prosperity and Collapse. Simulation of HandyEgal3.

The oscillatory mode exhibits “reversible collapses”. This scenario can be seen as belonging to a “limit cycle” for the values of the depletion factor. If `depletFactor` increases even further, the system changes into a different mode, one of “irreversible” collapse, with only one overshoot-and-collapse cycle. This is obtained in the scenario HandyEgal4.

Additional Scenarios in an Egalitarian Type of Society

There are also scenarios models HandyEgal2 “Oscillatory Approach to Equilibrium” and HandyEgal4 “Full

Collapse”, corresponding to `depletFactor = 2.5*depletFactorEq` and `depletFactor = 5.5*depletFactorEq`, respectively. The reader is encouraged to test these models.

4.4 Simulation Scenarios of an Equitable Society—Workers and Non-Workers with Equal access to Wealth

In this society, Non-Workers and Workers have equal access to wealth, i.e., `ineqFactor = 1`. The non-workers’ privilege is that they get access to wealth without having to work (in that sense, they are regarded as the Elites). The initial population of non-workers is `elite = 25`. As before, we look for an equilibrium situation where the depletion factor is set to its optimal value, which for this society is `depletFactor = depletFactorEq = 8.33E-6`. This produces the first scenario HandyEquit1 model shown further below, an adaptation of the HandyEquitableBase model which is inherited by all Equit scenario models.

```

partial model HandyEquitableBase
  "Equit Scenarios: Equitable society
  (with Workers and Non-Workers)"
  extends HandyBase(ineqFactor = 1,
    depletFactorEq = 0.00000833);
  initial equation
    commons = 100.0;
  end HandyEquitableBase;
    
```

Scenario HandyEquit1: “Soft Landing to Optimal Equilibrium”

```

model HandyEquit1
  "Equitable society Scenario Equit1: Soft
  Landing to Optimal Equilibrium"
  extends HandyEquitableBase;
  initial equation
    elite = 25;
  equation
    depletFactor = depletFactorEq;
  end HandyEquit1;
    
```

We simulate and plot:

```

plot({commons_, elites_, nature_,
  wealth_, carryingCap_, carryingCapMax_})
    
```




Figure 5. Equilibrium in the presence of workers (Commoners) and non-workers (Elites) with equal consumption per person. Simulation of HandyEquit1.

In Figure 5, the sum of worker and non-worker populations reaches the maximum carrying capacity. The optimal depletion factor in this society is bigger than that of an egalitarian society analyzed in the previous section. This is due to an increased amount of production required from the workers to sustain the non-workers.

Scenario HandyEquit2: “Oscillatory Approach to Equilibrium”

If the depletion factor is further increased to $\text{depletFactor} = 2.64 * \text{depletFactorEq}$ we obtain an oscillatory behavior. The results are shown in Figure 6

```

model HandyEquit2
  "Equitable society Scenario Equit2:
  Oscillatory Approach to Equilibrium"
  extends HandyEquitableBase;
  initial equation
    elite = 25;
  equation
    depletFactor = 2.64 * depletFactorEq;
end HandyEquit2;

```

We simulate and plot:

```

plot({commoners_, elites_, nature_,
wealth_, carryingCap_, carryingCapMax_})

```



Figure 6. Oscillatory convergence to equilibrium with tolerable overshoot. Simulation of HandyEquit2.

Now, the carrying capacity (`carryingCap`) is lower than the maximum carrying capacity, but the total population approaches a steady state value after an oscillatory phase.

Additional Scenarios in an Equitable Type of Society

There are also scenarios HandyEquit3 “Cycles of Prosperity, Overshoot and Collapse”, HandyEquit4 “Full Collapse”, and HandyEquit5 “Preventing a Full Collapse by Decreasing Average Depletion per Capita”.

The reader is encouraged to also try these models, which include experimenting with different values of the depletion factor, and in the case of HandyEquit5, increasing the ratio of non-workers to workers.

4.5 Simulation Scenarios of an Unequal Society—Elites and Commoners with Different Access to Wealth

The unequal society appears to be closer to the status of our current world. The inequality factor `ineqFactor` is made to vary from 10 to 100 in the unequal scenarios.

We will reproduce a pair of experiments intended to show the effects of changing birth rates and inequality as a means for moving from an unsustainable to a sustainable mode of behavior.

```

partial model HandyUnEquitableBase
  "Uneq Scenarios: Unequal Society (with
  Elite and Commoners)"
  extends HandyBase(ineqFactor = 100,
    depletFactorEq = 0.00000667,
    birthRComm = 0.03, birthRElite = 0.03);
end HandyUnEquitableBase;

```

Scenario HandyUneq2: “Full Collapse”

In this scenario the inequality factor is set very high: `ineqFactor = 100`. Also, the initial seed for the population of the Elite is set to `elite = 0.2` and a large depletion factor (including pollution effects) `depletFactor = 1E-4` is selected. The result is a full collapse with no recovery, as shown in Figure 7 below.

```

model HandyUneq2
  "UnEquitable society Scenario Uneq2:
  Type-II Collapse (Full Collapse)"
  extends HandyUnEquitableBase(ineqFactor
  = 100,
    depletFactorEq = 0.00000667,
    birthRComm = 0.03, birthRElite = 0.03);
  initial equation
    elite = 0.2;
    commoners = 100.0;
  equation
    depletFactor = 15 * depletFactorEq;
end HandyUneq2;

```

```

plot({commoners_, elites_, nature_,
wealth_, carryingCap_, carryingCapMax_})

```




Figure 7. Full collapse due to over-depletion and high levels of inequality (`ineqFactor=100`). Simulation of HandyUneq2.

As soon as the Commoners' population surpasses the carrying capacity, `wealth` starts to decline and never again recovers. Before disappearing, Elites remain insensitive to the wealth's fast decrease for a long period after the Commoners' population starts its massive decline. This is possible due to the unequal access to wealth, which is a hundred times larger than that of the Commoners.

Scenario HandyUneq3: "Soft Landing to Optimal Equilibrium"

In this scenario several parameters and initial values are changed with respect to the previous case.

- Inequality is reduced by a factor of ten, yielding `ineqFactor = 10`.
- The depletion factor is set to its equilibrium value, derived analytically: `depletFactor = depletFactorEq = 6.35E-6`, much lower than in the previous scenario.
- The initial values for the population are set as `commoners=10 000` and `elite=3000`.
- The birth rates are assumed to be controllable and selected as `birthRComm=0.065` and `birthRElite=0.02`.

The results are shown below:

```

model HandyUnEq3
  "Unequitable society UnEquit3: Soft
  Landing to Optimal Equilibrium"
  extends HandyUnequitableBase(ineqFactor
  = 10, depletFactorEq = 0.00000635,
  birthRComm = 0.065, birthRElite =
  0.02);
  initial equation
    elite = 3000.0;
    commoners = 10000.0;
  equation
    depletFactor = depletFactorEq;
  end HandyUnEq3;
plot({commoners_, elites_, nature_,
wealth_, carryingCap_, carryingCapMax_})

```




Figure 8. Unequal society reaching a sustainable equilibrium with moderate inequality (`ineqFactor=10`) and birth rate control. Simulation of `HandyUnEq3`.

The new set of parameters produces a sustainable equilibrium in an unequal society. It represents an example of implementation of policies that simultaneously limit inequality and allow birth rates to remain below critical levels. In this model inequality and birth rate are separate but simultaneous measures. In fact, in the real world high birth rates in poor societies are often a consequence of inequality.

Additional Scenarios for an Unequal Type of Society

There are also scenarios `HandyEquit1` "Collapse with Recovery of Nature" and `HandyEquit4` "Oscillatory Approach to Equilibrium", which perform experiments with different values of the depletion factor, birth rates, and values of initial populations.

5 A Short Look at the World3 Model

`World3` is a rather complex world model and therefore, its realm, scope, and objectives are quite different from those of `Handy`. The compiled version contains 41 state variables and 245 algebraic variables, with the same number of equations, representing many facets of human society and global ecological and economic dynamics.

The model is available as part of the `SystemDynamics.WorldDynamics` library [1], developed by François Cellier and his students. They made a documented Modelica version of `World3` by translating the original model from its version in Stella.

The `SystemDynamics.WorldDynamics.World3` model is represented as a Modelica package that implements Meadows et al `World3` model. `World3` was first described in the 1972 book *Limits to Growth: A Report for the Club of Rome's Project on the Predicament of Mankind* [5]. The book has seen two later editions, one published in 1992 (20-year edition), and the most recent one published in 2004 (30-year edition) [7]. Each of these new editions were accompanied by an updated model. The model in this Modelica `World3` version is the newest model discussed in the 2004 edition *Limits to Growth: The 30-Year Update*.

Whereas Jay Forrester listed his entire mathematical model in his book *World Dynamics*, Meadows et al in *Limits to Growth* only discussed the results obtained using their model. The mathematical model itself was not listed. The main reason was to make the book more accessible to a wider public. Another reason is that `World3` is considerably more complex than `World2`, and consequently, a thorough discussion of all aspects of the model would have taken up much more space in the book. Instead, the authors published a separate book of 637 pages in 1974: *Dynamics of Growth in a Finite World* [6] that describes all facets of the model.

Since the World3 model is rather complex, it was subdivided into 13 different sectors (i.e., submodels) describing aspects of the following: 1) arable land dynamics, 2) food production, 3) human ecological footprint, 4) human fertility, 5) human welfare index, 6) industrial investment, 7) labor utilization, 8) land fertility, 9) life expectancy, 10) non-recoverable resource utilization, 11) pollution dynamics, 12) population dynamics, and 13) service-sector investment.

In the overall main model, one submodel of each class is placed on the screen, and the individual submodels are connected appropriately (Figure 9).




Figure 9. Overall World3 model.

The submodel World3.Land_Fertility, for example, is depicted below in Figure 10.




Figure 10. A submodel of World3: SystemDynamics.World3.Land_Fertility.

This submodel has 3 Inputs and 2 outputs. Two state variables, Land_Fertility and Yield_Tech, are represented by means of “Level” blocks, which have the ability to accumulate. Levels play the role of the state variables in a differential equation. Levels are in turn driven by the input and/or output “Rates” connected to the left and right sides, respectively, determining the velocities at which the accumulated quantity increases

or decreases. These Rates, in turn can be functions of other variables, including the state of the Level itself. Said functions are defined and interconnected following a block diagram approach, resorting to a vast library of pre implemented blocks such as Non-Linear Functions, Table Lookup Functions, Gains, Multipliers, etc.

The “Clouds” connected to the rates have only a pictorial purpose, expressing the fact that a source and/or a sink always exists with the physical ability to provide and/or absorb material without limits. The latter can be considered the conceptual “boundaries” of the model, as no detail is given for the processes taking place in material supply or consumption at these ends.

Whereas the World2 model lumps the entire population into a single state variable, the World3 model offers a demographic population dynamics model that distinguishes between children and adolescents, young adults of child-bearing age, older adults who are still integrated into the work force, and the retired population.

The capital investment is subdivided into investments in the military/industrial complex, in the service sector, and in agriculture.

Both the natural resources and pollution models have been upgraded by including changes in technology as factors influencing the depletion of resources and the release of pollutants. This is meaningful as improved technology may enable us to use the available resources more efficiently, and may also make it possible to produce goods in a cleaner fashion.

Scenarios with World3

The book *Limits to Growth: The 30-Year Update*, describes eleven scenarios based on different sets of assumptions. These scenarios are also part of the Modelica World3 model. We shall briefly discuss three of these scenarios:

- Scenario1 is the so-called “Standard Run”. This is the original World3 model with basic assumptions and without any adjustments.
- Scenario2 is the same as Scenario1, but with twice as much non-recoverable resources initially available, i.e., more oil, coal, metal, etc.
- Scenario9 is based on Scenario2 but applies a number of measures to avoid collapse and to try to transform the human society into a sustainable one.

The three scenarios are simulated between year 1900 and 2200.

Unfortunately, both Scenario1 and Scenario2 lead to population collapse rather quickly, as depicted in Figure 11. Doubling the non-renewable resources in sce-

scenario2 does not help at all (it delays the collapse, but makes it steeper). The population increases a little further, but the subsequent collapse is even more severe, primarily due to increased levels of pollution. The current developments in our world unfortunately seem to be rather close to Scenario2.

If a collapse indeed should occur, will it be as quick as depicted in Figure 11? That must not necessarily be the case. The World3 model assumptions may not be completely valid during such a stressful transition period. Political decisions, rationing policies, etc., may lead to decreasing resource consumption and create a slower, more controlled contraction to a sustainable level of consumption.




Figure 11. The world population according to Scenarios 1, 2, and 9 in the World3 model.

How can a collapse be avoided? Scenario9 applies the following aggressive technology development plan:

- Increased industrial resource efficiency, i.e., a reduction in the use of non-renewable material and energy, a 4% resource decrease/year.
- Arable land protection, e.g., to decrease and prevent land erosion, and to preserve existing land.
- Agricultural yield enhancement, a 4% increase/year.
- Pollution reduction, a 4% decrease/year.

Additionally, a program for birth control is implemented with an average number of two children per family. The scenario furthermore assumes that there is a certain capital cost to implement these technologies, and that 20 years are required for their full implementation.

The Scenario9 simulation to year 2200 shows a sustainable society. However, if the simulation continues to the year 2500, we observe that it is hard to maintain sustainability with a large human population combined with high standards of living. If the size of the population is gradually decreased back to its level of about a hundred years ago, the chances for being able to maintain high standards of living are much better.

Another issue is model validity. A number of assumptions behind the World3 model may not be valid that far into the future. However, one fact is certain: the human race must, in the long run, live sustainably on planet Earth due to limitations in resources, energy, and space. Consumption overshoots like the current one can only be temporary.

6 Conclusion

The Handy model has the remarkable feature of providing a minimal structure helpful for intuitive understanding of the human-nature interaction. It can produce a rich variety of dynamical modes, leading to non-trivial scenarios. This combination is seldom found in most world models, where the quest for including more details comes at the price of lack of intuitiveness.

Handy is very useful as a “conceptual” model that can trigger varied interpretations and discussions comparing qualitatively different scenarios.

However, the goal of Handy is not to make quantitative short-term forecasts of the world state, but to assess qualitatively the impacts of various factors such as inequality and depletion on the long-term behavior of human societies.

The explicit stratification of social classes guides the interpretations to an essential playground needed to discuss the fate of human population.

Also worth mentioning is the fact that human societies developed a degree of cultural and technological complexity that prevents treating them in the same way as an animal species.

Accumulation of wealth (human’s ability to save throughout generations) is a major distinction which allows for irreversible population collapses, whereas with predation only, the predator-prey type of models can only produce cyclic oscillations (i.e., partial collapses).

Several senior modelers have remarked that it is not reasonable to expect that any world model will correctly predict how the world will develop after the onset of collapse, as unpredictable social disruptions and/or restructurings would result that invariably invalidate most of the model’s assumptions, parameters and even structures that might have described pre-collapse phases in a credible way. Such social reshaping will have its dynamics determined by interactions between different social classes, and in that regard, again, the Handy approach of disaggregating them in at least two strata represents a valuable contribution to future world modeling.

Moreover, we want to draw the reader’s attention to the aggregation made in Handy of natural renewable and non-renewable resources into a stock variable. This

paves the way for attaining scenarios that look sustainable in the very long run (i.e., when simulating “to infinity”). Based on the principles of thermodynamics it can be shown that in the long run the maximum carrying capacity of humans on earth will be determined *only* by the remaining renewable resources (once we have depleted all the non-renewables for good).

If we consider renewable and non-renewable resources independently of each other (along with the commoners, the elites, and their inequality factor), we may be able to use the resulting world model to investigate scenarios that offer possible answers to an additional highly relevant question. How should we best use the non-renewable resources to attain a society with minimum inequalities, for when the time comes that humanity must rely on renewable resources only? To investigate such a goal, the inequality factor should be altered from a constant parameter to a variable. Although models presented in this paper are from different classes (minimal Handy vs. more complex, realistic world model, World3), their conclusions are similar. In the long run, not so far into the future, humanity must change to living sustainably on planet Earth. This change can occur either as a planned gradual transition, preserving well-functioning societies, or as a more disruptive, unplanned transition resulting in a less pleasant society with a reduced ecological capacity.

7 Acknowledgments

This work has been supported by the Swedish Governmental Agency for Innovation Systems (Vinnova) within the ITEA2 MODRIO project, and by the Swedish Research Council (VR).

References

- [1] Cellier, François. World3 in Modelica: Creating System Dynamics Models in the Modelica Framework. In *Proceedings of the 6th International Modelica Conference*, Bielefeld, Germany. 2008.
- [2] Fritzson, Peter. *Principles of Object Oriented Modeling and Simulation with Modelica 2.1*, 940 pages, ISBN 0-471-471631, Wiley-IEEE Press. Feb. 2004.
- [3] Lotka, Alfred J. *Elements of Mathematical Biology*. Dover Publications, New York, 1956.
- [4] Malthus, T. *An Essay on the Principle of Population*. Printed for J. Johnson, in St. Paul’s Church-Yard, London, 1798. (Reprint. Amherst, NY: Prometheus Books, 1998).
- [5] Meadows D., Meadows D., Randers J., and Behrens III W.W. *Limits to Growth: A Report for the Club of Rome’s Project on the Predicament of Mankind*. 205p. Universe Books, New York, 1972.
- [6] Meadows D., Behrens III W.W., Meadows D., Naill R.F., Randers J., and Zahn E.K.O. *Dynamics of Growth in a Finite World*. 637p. Wright-Allen Press, 1974.
- [7] Meadows D., Randers J., and Meadows D. *Limits to Growth: The 30-Year Update*. 368p. Chelsea Green, 2004.
- [8] Modelica Association. Modelica—A Unified Object-Oriented Language for Physical Systems Modeling: Language Specification Version 3.2 rev 2. Available at <http://www.modelica.org>, August, 2013.
- [9] Motesharrei S., Rivas J., and Kalnay E. A Minimal Model for Human and Nature Interaction. Accepted for publication (expected 2014) *Journal Ecological Economics*.